Jの組み込み乱数

M.Shimura
JCD02773@nifty.ne.jp

2008 年 4 月 16 日

1　Jの組み込み乱数

classes/packages/stats/

random.ijs,statdist.ijs

Ver.6 で 乱数源にメルセンヌを採用し、強化されている。

New random number generators, including Mersenne Twister as the default RNG.

■樅の木の植樹　樅の木を一本植える。

分布のパラメータは次の 2 つがよく用いられる。

位置　location parameter

木の高さと樹形 scale parameter

1.1 一覧

一様乱数	rand01 v rand11	generate y random numbers in interval (0,1) generate y random numbers in interval (_1,1)	
binom	binomialdist binomialprob binomialrand	y has 2 elements p,n:	
normal	normalprob normalrand	y has 3 or 4 elements: 0 = mean of distribution 1 = standard deviation 2 = minimum result 3 = maximum result (rand) with mean=0, standard deviation=1.	Box Muller method
Γ	gammarand	y has 2 elements p,n 0 = power parameter 1 = number of trials if p=1 this is the exponential distribution	$f(x) = \frac{1}{\Gamma(\lambda)}\alpha^{\lambda}x^{\lambda-1}e^{-\alpha x}$
β	betarand	y has 3 elements p,q,n p is probability of success in one trial q is 1-p n is number of trials	$f(x) = \frac{1}{B(p,q)}x^{p-1}(1-x)^{q-1}$ $E(x) = \frac{p}{p+q}$ $V(x) = \frac{pq}{(p+q)^2(p+q+1)}$ $B(p.q) = \frac{\Gamma(p)\Gamma(q)}{\Gamma(p+q)} = \frac{(p-1)!(q-1)!}{(p+q+1)!}$

poisson	poissondist poissonprob poissonrand	y has 2 elements: 0 = mean of distribution 1 = maximum value to show e.g. poissondist 2 10 = list of probabilities of values from 0 to 10 in poisson distribution of mean 2	
exponential	exponentialrand	with mean=1. `F(x)=1-^-x` y = number of trials	
cauchy	cauchyrand	$F(x) = 0.5 + (\arctan x)\%o.1$ y is number of trials	

1.2 試運転

binomial random numbers

```
  binomialrand 0.4 20
1 1 1 1 1 0 1 1 1 0 0 1 1 1 0 1 1 0 0 0
```

```
binomialrand=: 3 : 0
'p n'=. y
r=. 2147483647
s=. <:p*r
s>?n#<:r
)
```

normal random numbers

Box Muller Method

互いに独立な一様乱数　U_1, U_2

$Z_1 = \sqrt{-2logU_1}cos(2\pi U_2)$

$Z_2 = \sqrt{-2logU_1}sin(2\pi U_2)$

```
normalrand=: 3 : 0
(2 o. +: o. rand01 y) * %: - +: ^. rand01 y
)
```

Z_1 is cos

without parameter

'stick' plot 100 hist_count normalrand 10000



図 1　normalrand

poisson random numbers

$f(x) = \dfrac{\lambda^x e^{-\lambda}}{x!}$

```
poissonrand=: 3 : 0
'm n'=. y
roll=. -@^.@rand01
r=. b=. m >: t=. roll n
i=. i.n
while. #i=. b#i do.
  b=. m >: t=. (b#t) + roll #i
  r=. (b + i{r) i } r
end.
r
)
```

'bar' plot 100 hist_count poissonrand 3 10000



図 2　*poisson*

exponential random numbers

$f(x) = \lambda e^{-\lambda x}$

```
exponentialrand=: 3 : 0
-^.rand01 y
)
```

$f(x) = e^{-\lambda x}$

'stick' plot 100 hist_count exponentialrand 10000



図 3　e

Γ *random numbers*

$$f(x) = \frac{1}{\beta^{\alpha}\Gamma(\alpha)} x^{\alpha} e(-\frac{x}{\beta})$$

gammarand(p,n)

p is $\lambda(=\alpha)$, β はなし）

p=1 is exponential rand /指数分布になる

独立な指数分布に従う確率変数の n 個の和である。

$\alpha = 1$ の場合は平均 β の指数分布 $(\lambda = \frac{1}{\beta})$

α が正の整数の場合はアーラン分布である。

$a = \frac{m}{2}, \beta = 2$ とすると自由度 m のχ^2 分布

```
gammarand=: 3 : 0
'p n'=. y
r=. n#0
k=. p-i=. <.p
if. k do.
  r=. betarand k,(-.k),n
  r=. r * -^.rand01 n
end.
if. i do.
  r-^.*/rand01 i,n
end.
)
```

'stick' plot 10 hist_count gammarand 2 10000



図 4　$\Gamma(\lambda = 2)$

'stick' plot 10 hist_count gammarand 3 10000



図 5　$\Gamma(\lambda = 3)$

β *randomnumbers*

$$f(x) = \frac{1}{B(a,b)} x^{a-1}(1-x)^{b-1}$$

$$where, B(a,b) = \int_0^1 t^{a-1}(1-t)^{b-1}dt$$

betarand(p1,p2,n)

p1 is λ_1, p2 is λ_2

```
betarand=: 3 : 0
'p q n'=. y
if. (1>p) *. 1>q do.
  b=. n#1
  r=. n#0
  whilst. 1 e. b do.
    m=. +/b
    x=. (rand01 m)^%p
    y=. x+(rand01 m)^%q
    t=. 1>:y
    z=. (t#x)%t#y
    i=. t#b#i.#b
    b=. 0 i } b
    r=. (z+i{r) i } r
  end.
else.
  s%(gammarand q,n)+s=. gammarand p,n
end.
)
```

'stick' plot 10 hist_count betarand 2 3 10000



図 6　$\beta(\lambda_1 = 2,\ \lambda_2 = 3)$

'stick' plot 10 hist_count betarand 4 6 10000



図 7　$\beta(\lambda_1 = 4,\ \lambda_2 = 6)$

cauchy random numbers

```
NB. random numbers in a cauchy distribution
NB. with F(x)=0.5+(arctan x)%o.1
NB.
NB. y  =  number of trials
cauchyrand=: 3 : 0
3 o. o. _0.5+rand01 y
)
```

'stick' plot 100 hist_count cauchyrand 10000



図 8　*cauchy*

```
 discreterand ((i.4);0.1 0.3 0.4 0.2);10
1 1 3 1 1 2 2 2 3 2
```

■Erlang　*Anger Krarup Erlang*(1878-1929) Denmark

コペンハーゲン大学出身、学校で数年間数学を中心に物理、天文、化学を研究し、教えた。数学協会でCTT(Copenhagen Telephone and Telegram) の主任技術者 *Jensen* から電話のコールと待ち時間の問題の解決を要請される。1908 年から CTT に在籍し、待ち行列の研究に従事。

1909　*The theory of probability and telephone conversation*

Erlang の待ち行列の理論は British Post Office を始めとし各国の電話会社で応用された。

対数などの数学関数の数表の作成にも業績を残した。

issue :(http://www-history.mcs.st-and.ac.uk/history/Biographies/Erlang.html/)